
Chapter 6

Systems of Linear Equations

6.1 Introduction

Systems of linear equations hold a special place in computational physics, chemistry, and engi-
neering. In fact, multiple computational problems in science and technology can be mathemat-
ically expressed as a linear system. Most methods in computational mathematics for solving
systems of partial differential equations (PDE), integral equations, and boundary value problems
in ordinary differential equations (ODE) utilize the Finite Difference Approximation, effectively
replacing the original differential and integral equations on systems of linear equation. Ad-
ditionally, other applications of systems of linear equations in numerical analysis include the
linearization of systems of simultaneous nonlinear equations, and the fitting of polynomials to
a set of data points.

A system of linear equations has the following form

a11x1 + a12x2 + a13x3 + . . . + a1nxn = b1

a21x1 + a22x2 + a23x3 + . . . + a2nxn = b2

. . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . .
an1x1 + an2x2 + an3x3 + . . . + annxn = bn

(6.1)

where xj(j = 1, 2, . . . , n) are unknown variables, aij(i, j = 1, 2, . . . , n) are the coefficients, and
bi(i = 1, 2, . . . , n) are the nonhomogeneous terms. The first subscript i identifies the row of
the equation and the second subscript j identifies the column of the system of equations. The
system (6.1) can also be written as a matrix equation⎛

⎜⎜⎜⎝
a11 a12 a13 . . . a1n

a21 a22 a23 . . . a2n

. . . . . . . . . . . . . . .

an1 an2 an3 . . . ann

⎞
⎟⎟⎟⎠
⎛
⎜⎜⎜⎝

x1

x2

. . .

xn

⎞
⎟⎟⎟⎠ =

⎛
⎜⎜⎜⎝

b1

b2

. . .

bn

⎞
⎟⎟⎟⎠ , (6.2)

or in a compact form as Ax = b

Methods for solving linear systems are normally taught in mathematical classes and courses
of linear algebra. The standard syllabus includes the substitution method, Cramer’s rule, and

the inverse matrix. Unfortunately, Cramer’s rule is a highly inefficient method for solving real
systems of linear equations: the number of equations in a system may run into the hundreds,
thousands, or even millions (e.g. structure and spectra calculations for quantum systems). Since
Cramer’s rule is based on evaluations of multiple determinants, it needs about n! multiplications
and divisions for solving a system of n equations. Thus, solving a system of only 30 equations
(30! ∼ 2 · 1032) would take as much time as the age of the universe on a teraflop computer.
Another common idea in standard linear algebra courses is that the solution to Ax = b can
be written as x = A−1b, where A−1 is the inverse matrix of A. However, in most practical
computational problems, it is not recommended to compute the inverse matrix to solve a system
of linear equations. In fact, it normally takes more operations to compute the actual inverse
matrix instead of simply finding the solution by one of the direct elimination methods. Finally,
the method of substitution, well known for high school students, is the foundation for multiple
methods in numerical analysis for solving real problems.

There are two classes of methods for solving systems of linear equations. In direct methods,
a finite number of arithmetic operations leads to an ”exact” (within round-off errors) solution.
Examples of such direct methods include Gauss elimination, Gauss-Jordan elimination, the
matrix inverse method, and LU factorization. The average number of operations to solve a
system of linear equations for these methods is ∼ n3. Iterative methods achieve the solution
asymptotically by an iterative procedure, starting from the trial solution. The calculations
continue until the accepted tolerance ε is achieved. Jacobi, Gauss-Seidel, and successive over-
relaxation, are all examples of iterative methods. Direct elimination methods are normally used
when the number of equations is less than a few hundred, or if the system of equations is ill-
conditioned. Iterative methods are more common for large and diagonally dominant systems of
equations, especially when many non-diagonal coefficients equal zero or very small numbers.

At present, there are multiple algorithms and programs developed for solving systems of
linear equations based on direct and iterative methods. Using a method that utilizes the most
from the matrix shape (symmetric, sparse, tridiagonal, banded) results in higher efficiency and
accuracy. The most common problems in matrix calculations are the results of round-off errors
or the running out of memory and computational time for large systems of equations. It is
also important to remember that various computer languages may handle the same data very
differently. For example, in C/C++, the first element of an array starts from index 0, in Fortran
(by default), from index 1. It is also useful to note that Fortran 90/95 has numerous intrinsic
functions to do matrix calculations.

In this chapter, we will consider linear systems (6.1) to have real coefficients aij. We will also
assume an existence of a unique solution (e.g. det A �= 0 if the right-hand coefficients bi �= 0, or
det A = 0 if bi = 0).

Comments: 1) possible examples from physics: electric circuits, equilibrium problems

6.2 Direct elimination methods

Elimination methods use a simple idea that is well known from courses of algebra: a system of
two equations worked out formally by solving one of the equations. Let’s say we solve the first
equation for the unknown x1 in terms of the other unknown x2. Substituting the solution for x1

into the second equation gives us a single equation for one unknown x2, thus x1 is eliminated
from the second equation. After x2 is found, the other x1 unknown can be found by back
substitution.

In the general case of n linear equations, the elimination process employs operations on rows
of linear equations that do not change the solution, namely, ”scaling” - any equation may be
multiplied by a constant, ”pivoting” - the order of equations can be interchanged, ”elimination”
- any equation can be replaced by a linear combination of that equation with any other equation.

6.2.1 Basic elimination

For a good understanding of basic techniques of direct elimination, it is incredibly helpful to
apply the elimination method to find solutions of a system of three linear equations

a11x1 + a12x2 + a13x3 = b1

a21x1 + a22x2 + a23x3 = b2 (6.3)

a31x1 + a32x2 + a33x3 = b3

Step 1a: Subtracting the first equation multiplied by a21/a11 from the second equation, and
multiplied by a31/a11 from the third equation gives

a11x1 + a12x2 + a13x3 = b1

(a21 − a21)x1 + (a22 − a21
a11

a12)x2 + (a23 − a21
a11

a13)x3 = b2 − a21
a11

b1

(a31 − a31)x1 + (a32 − a31
a11

a12)x2 + (a33 − a31
a11

a13)x3 = b3 − a31
a11

b1

(6.4)

One can see that the coefficients by the unknown x1 in the second and the third rows of the new
system are zero

a11x1 + a12x2 + a13x3 = b1

0 + a′22x2 + a′23x3 = b′2
0 + a′32x2 + a′33x3 = b′3

(6.5)

where a′ij = aij − ai1a1j

a11
and b′i = bi − ai1

a11
b1. Thus, we eliminated the first unknown x1 from the

second and third equations.
Step 1b: Now, let’s subtract the modified second equation multiplied by a′32/a′22 from the third
equation in (6.5)

0 + (a′32 − a′32)x2 + (a′33 − a′23

a′32

a′22

)x3 = b′3 − b′2
a′32

a′22

(6.6)

After the two eliminations we have a new form for the system (6.3)

a11x1 + a12x2 + a13x3 = b1

0 + a′22x2 + a′23x3 = b′2
0 + 0 + a′′33x3 = b′′3

(6.7)

with a′′ij = a′ij −
a′

i2a′
2j

a′
22

and b′′i = b′i − a′
i2

a′
22

b2. Thus, the original system Ax = b is reduced to
triangular form.
Step 2: The last equation in (6.7) can be solved for x3, and the second for x2, and finally the
first for x1

x3 = b′′3/a′′33

x2 = (b′2 − a′23x3)/a′22 (6.8)

x1 = (b1 − a12x2 − a13x3)/a11

The basic elimination method can be easily generalized for a general n by n system, Ax = b

Algorithm 6.1 The basic elimination algorithm for solving a system of n linear equations.
Step 1: Apply the elimination procedure to every column k (k = 1, 2, . . . , n − 1) for rows i

(i = k + 1, k + 2, . . . , n) to create zeros in column k below the pivot element ak,k

ai,j = ai,j − (ai,k/ak,k) ak,j (i, j = k + 1, k + 2, . . . , n) (6.9)

bi = bi − (ai,k/ak,k) bk (i, j = k + 1, k + 2, . . . , n) (6.10)

Step 2: The solutions of the reduced triangular system can then be found using the backward
substitution

xn = (bn/an,n) (6.11)

xj =
1

ai,i

⎛
⎝bi −

n∑
j=i+1

ai.jxj

⎞
⎠ (i = n − 1, n− 2, . . . , 1) (6.12)

The total number of multiplications and divisions done by the basic elimination algorithm for
a system of n equations is about O(n3). The back substitution takes approximately O(n2)
multiplication and divisions.

Comments: Every next elimination uses results from the elimination before. For large sys-
tems of equations the round-off errors may quickly accumulate. Say again that it takes finite
number of steps to get a true (within the round-off error solution)

The program below implements the basic elimination for a general n by n matrix A

Program 6.1 The basic elimination.

subroutine gauss_1(a,b,x,n)
!============================================================
! Solutions to a system of linear equations A*x=b
! Method: the basic elimination (simple Gauss elimination)
! Alex G. November 2009
!-----------------------------------------------------------
! input ...
! a(n,n) - array of coefficients for matrix A
! b(n) - vector of the right hand coefficients b
! n - number of equations
! output ...
! x(n) - solutions
! comments ...
! the original arrays a(n,n) and b(n) will be destroyed
! during the calculation
!===========================================================
implicit none
integer n
double precision a(n,n), b(n), x(n)
double precision c
integer i, j, k

!step 1: forward elimination
do k=1, n-1

do i=k+1,n
c=a(i,k)/a(k,k)
a(i,k) = 0.0
b(i)=b(i)- c*b(k)
do j=k+1,n

a(i,j) = a(i,j)-c*a(k,j)
end do

end do
end do

!step 2: back substitution
x(n) = b(n)/a(n,n)
do i=n-1,1,-1

c=0.0
do j=i+1,n
c= c + a(i,j)*x(j)

end do
x(i) = (b(i)- c)/a(i,i)

end do
end subroutine gauss_1

Example 6.1 Solution by the basic elimination.

Basic elimination (Simple Gauss)

Matrix A and vector b
3.000000 2.000000 4.000000 4.000000
2.000000 -3.000000 1.000000 2.000000
1.000000 1.000000 2.000000 3.000000

Matrix A and vector b after elimination
3.000000 2.000000 4.000000 4.000000
0.000000 -4.333333 -1.666667 -0.666667
0.000000 0.000000 0.538462 1.615385

Solutions x(n)
-2.000000 -1.000000 3.000000

6.2.2 Gauss elimination

The first immediate problem with the basic elimination method comes when one of diagonal
elements is zero. For example, the following system

0x1 + 1x1 + 2x1 = 4

2x1 + 1x2 + 4x3 = 3 (6.13)

2x1 + 4x2 + 6x3 = 7

has a unique solution of x = {−2.5, 0.0, 2.0}. However, basic elimination would fail on the
first step since the a11 pivot element is zero. The procedure also fails when any of subsequent
ak,k pivot elements during the elimination procedure are zero. However, the basic elimination
procedure can be modified to push zero ak,k elements off the major diagonal. The order of
equations in a linear system can be interchanged without changing the solution. This procedure
is called ”partial pivoting”. ”Full pivoting” includes interchanging both equations and variables,
and it is rarely applied in practical calculations because of its complexity. Nevertheless, pivoting
can remove divisions by zero during the elimination process.

The effect of round-off errors can be reduced by scaling before pivoting. Scaling selects
an equation with the relatively largest pivot element akk. On every step k of the elimination
procedure we a) look first for a largest element ai,j in every row i = k, k + 1, . . . , n and scale
(normalize) every element in that row on the largest element, b) look for the largest element
ai,k in the column k to have it as a pivot element for the next elimination, c) interchange the
current equation k with the equation with the largest pivot element.

Let’s apply scaled pivoting to the system (6.13). The first scaling gives the following ai,1

elements {0.00, 0.50, 0.33}. Therefore, we rearrange the system placing the second equation as

the first one, and the third equation into second place.

2x1 + 1x2 + 4x3 = 3

2x1 + 4x2 + 6x3 = 7 (6.14)

0x1 + 1x1 + 2x1 = 4

After the first elimination, we have

2x1 + 1x2 + 4x3 = 3

0x1 + 3x2 + 2x3 = 4 (6.15)

0x1 + 1x2 + 2x3 = 4

The second scaling gives {1.00, 0.50} for ai,2 elements where i ≥ 2. Therefore, we keep the same
order of equations. After the second elimination, the original matrix is transformed to the upper
triangular form

2.00x1 + 1.00x2 + 4.00x3 = 3.00

0.00x1 + 3.00x2 + 2.00x3 = 4.00 (6.16)

0.00x1 + 0.00x2 + 1.33x3 = 2.66

The backward substitution returns the solutions {2.0, 0.0,−2.5}
The Gauss elimination includes all three basic operations on rows of linear equations: scaling,

pivoting and elimination.

Algorithm 6.2 Gauss elimination for solving a system of n linear equations.
Step 1: Apply the scaling, pivoting and elimination to every column k (k = 1, 2, . . . , n − 1)
starting from k = 1
a). Find the largest element in every row i = k, k + 1, . . . , n and divide other elements of those
rows by the corresponding largest element.
b). Find the largest pivoting element ai,k in a given column k for i = k, k + 1, . . . , n. Let’s say
it was al,k

c). Interchange rows k and l to have the relatively largest akk into the pivot position.
d). Apply the elimination procedure to the column k for rows i (i = k + 1, k + 2, . . . , n)

ai,j = ai,j − (ai,k/ak,k) ak,j (i, j = k + 1, k + 2, . . . , n) (6.17)

bi = bi − (ai,k/ak,k) bk (i, j = k + 1, k + 2, . . . , n) (6.18)

Step 2. Now it is time for backward substitution. At this point all the diagonal elements are
non zero, if the matrix is not singular. From the last equation we have

xn = (bn/an,n) (6.19)

Solving the other unknowns in the reverse order

xj =
1

ai,i

⎛
⎝bi −

n∑
j=i+1

ai.jxj

⎞
⎠ (i = n − 1, n− 2, . . . , 1) (6.20)

The solution is achieved in a finite number of steps determined by the size of the system. The
partial pivoting takes a very small fraction of computational efforts comparing to the elimination
calculations. The total number of operations is about O(n3). If all the potential pivots elements
are zero, then the matrix A is singular. Linear systems with singular matrices either have no
solutions, or do not have a unique solution.

Program 6.2 Gauss elimination with scaling and pivoting.

subroutine gauss_2(a,b,x,n)
!===========================================================
! Solutions to a system of linear equations A*x=b
! Method: Gauss elimination (with scaling and pivoting)
! Alex G. (November 2009)
!-----------------------------------------------------------
! input ...
! a(n,n) - array of coefficients for matrix A
! b(n) - array of the right hand coefficients b
! n - number of equations (size of matrix A)
! output ...
! x(n) - solutions
! coments ...
! the original arrays a(n,n) and b(n) will be destroyed
! during the calculation
!===========================================================
implicit none
integer n
double precision a(n,n), b(n), x(n)
double precision s(n)
double precision c, pivot, store
integer i, j, k, l

! step 1: begin forward elimination
do k=1, n-1

! step 2: "scaling"
! s(i) will have the largest element from row i

do i=k,n ! loop over rows
s(i) = 0.0

do j=k,n ! loop over elements of row i
s(i) = max(s(i),abs(a(i,j)))

end do
end do

! step 3: "pivoting 1"
! find a row with the largest pivoting element

pivot = abs(a(k,k)/s(k))
l = k
do j=k+1,n

if(abs(a(j,k)/s(j)) > pivot) then
pivot = abs(a(j,k)/s(j))
l = j

end if
end do

! Check if the system has a sigular matrix
if(pivot == 0.0) then

write(*,*) ’ The matrix is sigular ’
return

end if

! step 4: "pivoting 2" interchange rows k and l (if needed)
if (l /= k) then

do j=k,n
store = a(k,j)
a(k,j) = a(l,j)
a(l,j) = store

end do
store = b(k)
b(k) = b(l)
b(l) = store

end if

! step 5: the elimination (after scaling and pivoting)
do i=k+1,n

c=a(i,k)/a(k,k)
a(i,k) = 0.0
b(i)=b(i)- c*b(k)
do j=k+1,n

a(i,j) = a(i,j)-c*a(k,j)
end do

end do
end do

! step 6: back substiturion
x(n) = b(n)/a(n,n)
do i=n-1,1,-1

c=0.0
do j=i+1,n
c= c + a(i,j)*x(j)

end do
x(i) = (b(i)- c)/a(i,i)

end do

end subroutine gauss_2

Example 6.2 Solution by Gauss elimination.

Gauss elimination with scaling and pivoting

Matrix A and vector b
0.000000 1.000000 2.000000 4.000000
2.000000 1.000000 4.000000 3.000000
2.000000 4.000000 6.000000 7.000000

Matrix A and vector b after elimination
2.000000 1.000000 4.000000 3.000000
0.000000 3.000000 2.000000 4.000000
0.000000 0.000000 1.333333 2.666667

Solutions x(n)
-2.500000 0.000000 2.000000

It is useful to remember that there are variations of the Gauss elimination. For example,
the Gauss-Jordan elimination transforms the matrix A to a diagonal form, with a subsequent
reduction to the identity matrix I . As a result, the transformed vector b is a solution vector.
Despite the fact that this method needs more computational time, it can be used to evaluate the
inverse of matrix A−1, so that AA−1 = I . On the other hand, LU factorization is very efficient
for solving multiple systems with the same matrix A but with different vectors b. The Thomas
algorithm treats tridiagonal systems of equations.

6.2.3 Computing inverse matrices and determinants

The inverse matrix A−1 can be computed using the same Gauss elimination procedure. Finding
an inverse matrix is equivalent to finding matrix X such as

AX = I (6.21)

This equation can be rewritten as

n∑
k=1

ai,kxk,j = δi,j (i, j = 1, 2, . . . , n), (6.22)

where δi,j is the Kronecker delta. Then the system (6.22) is actually a set of n independent
systems of equations with the same matrix A but different vectors b. Let’s define the two
following vectors

x(j) = {xi,j}, e(j) = {δi,j}, (i = 1, 2, . . . , n) (6.23)

Now the the j-th column of the inverse matrix A−1 is the solution of the linear system

Ax(j) = e(j) (j = 1, 2, . . . , n) (6.24)

The set of systems (6.24) can be solved with Gauss elimination. It is clear that finding the
inverse matrix requires n-times more computational time than the elimination procedure.

For the illustration of this method, we consider a system of three equations. The first column
of the inverse matrix X can be found from the following systems⎛

⎜⎝ a11 a12 a13

a21 a22 a23

a31 a32 a33

⎞
⎟⎠
⎛
⎜⎝ x11

x21

x31

⎞
⎟⎠ =

⎛
⎜⎝ 1

0
0

⎞
⎟⎠ . (6.25)

The next two columns of the inverse matrix X correspond to solutions of the linear equations
with the same matrix A and the right sides as

b =

⎛
⎜⎝ 0

1
0

⎞
⎟⎠ for the second column, and b =

⎛
⎜⎝ 0

0
1

⎞
⎟⎠ for the third column. (6.26)

Technically, we may use Gauss elimination algorithm for solving n systems of n linear equations
to find the inverse matrix. However, computationally, it is time consuming since we have to do
the elimination for the same matrix A over and over again.

On the other hand, the LU factorization algorithm is incredibly efficient for solving multiple
linear equations with the same matrix but different right-hand vectors b. Any matrix can be
written as a product of two other matrices, in particular as A = LU , where L and U are the lower
triangular and upper triangular matrices. If the elements on the major diagonal of L are equal
to one, the method is called the Doolittle method. For unity elements on the major diagonal of
U , the method is called the Crout method. For A = LU , the linear system of equations Ax = b

becomes LUx = b. Multiplying both sides of the system by L−1 gives L−1LUx = L−1b, and
then Ux = L−1b = d, where d is a solution of Ld = b. Now it should be easy to see that the
following algorithm would lead to a solution of the linear system. First, we calculate U and L

matrices using the Gaussian elimination. While getting U is the goal of the elimination, the L

matrix consists of the elimination multipliers with unity elements of the main diagonal (it would
correspond to the Dolittle method). For every vector b we solve Ld = b to find d, namely

di = bi −
i−1∑
k=1

li,kdk (i = 2, 3, . . . , n), note that d1 = b1 (6.27)

Then, xn = dn/U(n, n), and other solutions for the linear system Ux = d are

xi = di −
n∑

k=i+1

ui,kxk/ui,i (i = n − 1, n − 2, . . . , 1) (6.28)

Since the number of multiplications to find solutions from the last two equations are of the order
O(n2), we can see that the LU decomposition method is exceptionally helpful for computing
inverse matrices.

Program 6.3 Compute Inverse matrix using LU Doolittle factorization

subroutine inverse(a,c,n)
!============================================================
! Inverse matrix
! Method: Based on Doolittle LU factorization for Ax=b
! Alex G. December 2009
!-----------------------------------------------------------
! input ...
! a(n,n) - array of coefficients for matrix A
! n - dimension
! output ...
! c(n,n) - inverse matrix of A
! comments ...
! the original matrix a(n,n) will be destroyed
! during the calculation
!===========================================================
implicit none
integer n
double precision a(n,n), c(n,n)
double precision L(n,n), U(n,n), b(n), d(n), x(n)
double precision coeff
integer i, j, k

! step 0: initialization for matrices L and U and b
! Fortran 90/95 aloows such operations on matrices
L=0.0
U=0.0

b=0.0

! step 1: forward elimination
do k=1, n-1

do i=k+1,n
coeff=a(i,k)/a(k,k)
L(i,k) = coeff
do j=k+1,n

a(i,j) = a(i,j)-coeff*a(k,j)
end do

end do
end do

! Step 2: prepare L and U matrices
! L matrix is a matrix of the elimination coefficient
! + the diagonal elements are 1.0
do i=1,n

L(i,i) = 1.0
end do
! U matrix is the upper triangular part of A
do j=1,n

do i=1,j
U(i,j) = a(i,j)

end do
end do

! Step 3: compute columns of the inverse matrix C
do k=1,n

b(k)=1.0
d(1) = b(1)

! Step 3a: Solve Ld=b using the forward substitution
do i=2,n

d(i)=b(i)
do j=1,i-1
d(i) = d(i) - L(i,j)*d(j)

end do
end do

! Step 3b: Solve Ux=d using the back substitution
x(n)=d(n)/U(n,n)
do i = n-1,1,-1

x(i) = d(i)
do j=n,i+1,-1
x(i)=x(i)-U(i,j)*x(j)

end do

x(i) = x(i)/u(i,i)
end do

! Step 3c: fill the solutions x(n) into column k of C
do i=1,n

c(i,k) = x(i)
end do
b(k)=0.0

end do
end subroutine inverse

Example 6.3 Inverse matrix

Computing Inverse matrix

Matrix A
3.000000 2.000000 4.000000
2.000000 -3.000000 1.000000
1.000000 1.000000 2.000000

Inverse matrix A^{-1}
1.000000 0.000000 -2.000000
0.428571 -0.285714 -0.714286

-0.714286 0.142857 1.857143

The elimination method can be easily applied to compute matrix determinants. At the end
of the elimination procedure, the original matrix A is transformed to the upper triangular form.
For such matrices, the determinant is a product of diagonal elements.

det(A) = ±
n∏

i=1

aii = a11a22a33 . . . ann, (6.29)

where the sign depends on the number of interchanges. Let’s remember that pivoting changes the
value of the determinant (interchanging any two equations changes the sign of the determinant).
However, counting the number of equation interchanges would give us the proper sign for the
determinant.

6.2.4 Tridiagonal systems

When a system of linear equations has a special shape (symmetric, or tridiagonal), then it is
recommended to use a method specifically developed for this kind of equation. Such methods
are not only more efficient in term of computational time and computer memory, but also
accumulate smaller round-off errors.

Here is an example of a tridiagonal system of five equations⎛
⎜⎜⎜⎜⎜⎝

a11 a12 0 0 0
a21 a22 a23 0 0
0 a32 a33 a34 0
0 0 a43 a44 a45

0 0 0 a54 a55

⎞
⎟⎟⎟⎟⎟⎠

⎛
⎜⎜⎜⎜⎜⎝

x1

x2

x3

x4

x5

⎞
⎟⎟⎟⎟⎟⎠ =

⎛
⎜⎜⎜⎜⎜⎝

b1

b2

b3

b4

b5

⎞
⎟⎟⎟⎟⎟⎠ , (6.30)

It is clear to see that one only element is to be eliminated in every row, namely ai−1,i,
affecting only the diagonal elements and the right hand vector. Subsequently, the elimination
procedure for a tridiagonal matrix

ai,i = ai,i − (ai,i−1/ai−1,i−1) ai−1,i (i = 2, . . . , n) (6.31)

and

bi = bi − (ai,i−1/ai−1,i−1) bi−1 (i = 2, . . . , n) (6.32)

However, it is possible to improve the efficiency of this method even further. Instead of storing
all n×n elements of the matrix A, since there is no need to keep the zero elements, we may use
a smaller matrix such n × 3:⎛

⎜⎜⎜⎜⎜⎜⎜⎝

− c12 c13

c21 c22 c23

c31 a32 c33

. . . . . . . . .

cn−1,1 cn−1,2 cn−1,3

cn,1 cn,2 −

⎞
⎟⎟⎟⎟⎟⎟⎟⎠

(6.33)

where the coefficients cij are related to the coefficients of the original matrix A as

ci,1 = ai,i−1, ci,2 = ai,i, and ci,3 = ai,i+1. (6.34)

Then the elimination procedure for the new matrix C

ci,2 = ci,2 − (ci,1/ci−1,2) ci−1,3 (i = 2, 3, . . . , n) (6.35)

and

bi = bi − (ci,1/ci−1,2) bi,i−1 (i = 2, 3, . . . , n) (6.36)

After the forward elimination, the back substitution gives the solutions of the tridiagonal system

xn = bn/cn,2 (6.37)

xi = (bi − ci,3xi+1)/ci,2 (i = n − 1, n− 2, . . . , 1) (6.38)

This algorithm for solving tridiagonal systems is called the Thomas algorithm. Thus algorithm
is widely used in solving 3-point partial and ordinary differential equations (more details?)

Program 6.4 the Thomas method for tridiagonal systems

subroutine thomas(c,b,x,n)
!============================================================
! Solutions to a system of tridiagonal linear equations C*x=b
! Method: the Thomas method
! Alex G. November 2009
!-----------------------------------------------------------
! input ...
! c(n,3) - array of coefficients for matrix C
! b(n) - vector of the right hand coefficients b
! n - number of equations
! output ...
! x(n) - solutions
! comments ...
! the original arrays c(n,3) and b(n) will be destroyed
! during the calculation
!===========================================================
implicit none
integer n
double precision c(n,3), b(n), x(n)
double precision coeff
integer i

!step 1: forward elimination
do i=2,n

coeff=c(i,1)/c(i-1,2)
c(i,2)=c(i,2)-coeff*c(i-1,3)
b(i)=b(i)-c(i,1)*b(i-1)

end do

!step 2: back substitution
x(n) = b(n)/c(n,2)
do i=n-1,1,-1

x(i) = (b(i)- c(i,3)*x(i+1))/c(i,2)
end do
end subroutine thomas

Example 6.4 Solution by the Thomas method

The Thomas method for tridiagonal systems

Matrix A and vector b
0.000000 4.000000 -1.000000 0.000000

-1.000000 4.000000 -1.000000 0.000000

-1.000000 4.000000 -1.000000 0.000000
-1.000000 4.000000 -1.000000 0.000000
-1.000000 4.000000 -1.000000 0.000000
-1.000000 4.000000 -1.000000 0.000000
-1.000000 4.000000 -1.000000 0.000000
-1.000000 4.000000 0.000000 16.000000

Solutions x(n)
0.000395 0.001578 0.005919 0.022099
0.082476 0.307806 1.148748 4.287187

Pivoting destroys the tridiagonality, and cannot be used ... (more? ) However, as a rule,
tridiagonal systems representing real physical systems are diagonally dominant, and pivoting is
unnecessary. more ... the number of multiplicative operations ∼ 5n , that makes it much more
efficient comparing to Gauss elimination by a factor of ∼ n2.

6.2.5 Round-off errors and ill-conditioned systems

In the elimination methods, each elimination step uses results from the step before. For linear
systems with large numbers of equations, the round-off errors may strongly affect the solution.
Round-off errors can be minimized by using double precision calculations and scaled pivoting.
Therefore, for matrix calculations, it is vital to use high precision arithmetic. Unfortunately,
it takes additional computational resources (memory and time), but it is better then having
unreliable solutions.

The effect of round-off errors is especially dangerous for ill-conditioned systems, when doing
”everything right”, you may in fact get ”everything wrong”. Ill-conditioned systems are very
sensitive to small variations in the equation coefficient. There are no methods for solving this
problem other than increasing precision. If we cannot fix the problem, it is at least good to
know if we are dealing with an ill-conditioned system. The Ill-conditioned system has a matrix
similar to a singular form, and their determinant is close to zero. A commonly used measure of
the condition of a matrix is its condition number. In fact, the norm of a matrix can be used to
evaluate the condition number: there are several ways to define the norm of a matrix, but the
most widely accepted is the Euclidean norm

‖A‖ =

⎛
⎝ n∑

i=1

n∑
j=1

a2
i,j

⎞
⎠

1/2

. (6.39)

For a matrix equation Ax = b it follows from the norm definition that

‖A‖‖x‖ ≥ ‖b‖. (6.40)

A small change in the right-hand vector b results in a change in the solution x as

A(x + δx) = b + δb, (6.41)

or subtracting the original equation for this one

Aδx = δb or δx = a−1δb (6.42)

Using norm’s properties we may write

‖δx‖ ≤ ‖A−1‖‖δb‖ (6.43)

Combining together equations (6.40) and (6.43)

‖b‖‖δx‖ ≤ ‖A‖‖x‖‖A−1‖‖δb‖ (6.44)

or

|δx‖
‖x‖ ≤ ‖A‖‖A−1‖|δb‖‖b‖ = C(A)

|δb‖
‖b‖ , (6.45)

where the product of two norms

C(A) = ‖A‖‖A−1‖ (6.46)

is the condition number of matrix A. The condition number is always ≥ 1. Logically, the
condition number is a factor by which a small variation in the coefficients is amplified during
the elimination procedure. Since computing the inverse matrix takes more time than solving the
system itself, it is common to use estimations for ‖A−1‖ without actually calculating the inverse
matrix. The most sophisticated codes in numerical libraries estimate the condition number
along with the solution, giving users an idea about the accuracy of the returned result. For
ill-conditioned systems it is advisable to check the final solution by a direct substitution in the
original equation.

Here is an example. Consider the equation

3.000000x1 + 2.00x2 + 4.000000x3 = 4.00

3.000001x1 + 2.00x2 + 4.000002x3 = 4.00 (6.47)

1.000000x1 + 1.00x2 + 2.000000x3 = 3.00

The condition number of the matrixA is 1.3264 · 107. The single precision solution by the basic
elimination is x = {−2.000, 2.750, 1.125}, and the double precision solution is x = {−2.0, 3.0, 1.0}
(that is the true solution).

6.3 Iterative methods

Iterative methods cannot compete with direct elimination methods for arbitrary matrix A.
However, in certain types of problems, systems of linear equations have many ai,j elements as
zero, or close to zero (sparse systems). Under those circumstances, iterative methods can be

extremely fast. Iterative methods are also efficient for solving Partial Differential Equations by
finite difference or finite element methods.

The idea of the iterative solution of a linear system is based on assuming an initial (trial)
solution that can be used to generate an improved solution. The procedure is repeated until
convergence with an accepted accuracy solution occurs. However, for an iterative method to
succeed/converge, the linear system of equations needs to be diagonally dominant.

|ai,i| >
∑
j �=i

|ai,j|. (6.48)

Iterative methods are less sensitive to round-off errors in comparison to direct elimination meth-
ods.

Let’s consider a system of linear equations

n∑
j=1

ai,jxj = bi (i = 1, 2, . . . , n). (6.49)

Every equation can be formally solved for a diagonal element

xi =
1

ai,i

⎛
⎝bi −

i−1∑
j=1

ai,jxj −
n∑

j=i+1

ai,jxj

⎞
⎠ (i = 1, 2, . . . , n). (6.50)

Choosing an initial solution we may calculate the next iteration

xk+1
i =

1
ai,i

⎛
⎝bi −

i−1∑
j=1

ai,jx
k
j −

n∑
j=i+1

ai,jx
k
j

⎞
⎠ (i = 1, 2, . . . , n). (6.51)

Equation (6.51) can be rewritten in the iterative form

xk+1
i = xk

i +
1

ai,i

⎛
⎝bi −

n∑
j=1

ai,jx
k
j

⎞
⎠ (i = 1, 2, . . . , n). (6.52)

Equation (6.51) defines the Jacobi iterative method, which is also called the method of si-
multaneous iterations. It is possible to prove that if A is diagonally dominant, then the Jacobi
iteration will converge. The number of iterations is either predetermined by a maximum number
of allowed iterations, or by one of conditions for absolute errors

max
1≤i≤n

∣∣∣xk+1
i − xk

i

∣∣∣ ≤ ε, or
n∑

i=1

∣∣∣xk+1
i − xk

i

∣∣∣ ≤ ε, or

(
n∑

i=1

(
xk+1

i − xk
i

)2
)1/2

≤ ε, (6.53)

where ε is a tolerance. It is also possible to use another condition

‖Axk − b‖
‖b‖ < ε (6.54)

Since efforts to evaluate the norms above are comparable with the iterative calculations, it is
recommended to check the convergence based on equation (6.54) after every tenth iteration.

In the Jacobi method, all values of xk+1 are calculated using xk values. In the Gauss-Seidel
method, the most recently computed values of xi are used in calculations for j > i solutions

xk+1
i =

1
ai,i

⎛
⎝bi −

i−1∑
j=1

ai,jx
k+1
j −

n∑
j=i+1

ai,jx
k
j

⎞
⎠ (i = 1, 2, . . . , n), (6.55)

or

xk+1
i = xk

i +
1

ai,i

⎛
⎝bi −

i−1∑
j=1

ai,jx
k+1
j −

n∑
j=i

ai,jx
k
j

⎞
⎠ (i = 1, 2, . . . , n). (6.56)

The Gauss-Seidel iterations generally converge faster than Jacobi iterations.
Quite often, the iterative solution to a linear system approaches the true solution in the same

direction. Then it is possible to accelerate the iterative process by introducing the over-relaxing
factor ω

xk+1
i = xk

i + ω
1

ai,i

⎛
⎝bi −

i−1∑
j=1

ai,jx
k+1
j −

n∑
j=i

ai,jx
k
j

⎞
⎠ (i = 1, 2, . . . , n). (6.57)

For ω = 1 the system (6.57) is the Gauss-Seidel method, for 1.0 < ω < 2.0 the system is over-
relaxed, and for ω < 1.0 the system is under-relaxed. The optimum value of ω depends on the
size of the system and the nature of the equations. The iterative process (6.57) is called the
successive-over-relaxation (SOR) method.

Program 6.5 Gauss-Seidel: The successive-over-relaxation

subroutine gs_sor(a,b,x,omega,eps,n,iter)
!==========================================================
! Solutions to a system of linear equations A*x=b
! Method: The successive-over-relaxation (SOR)
! Alex G. (November 2009)
!----------------------------------------------------------
! input ...
! a(n,n) - array of coefficients for matrix A
! b(n) - array of the right hand coefficients b
! x(n) - solutions (initial guess)
! n - number of equations (size of matrix A)
! omega - the over-ralaxation factor
! eps - convergence tolerance
! output ...
! x(n) - solutions

! iter - number of iterations to achieve the tolerance
! coments ...
! kmax - max number of allowed iterations
!==========================================================
implicit none
integer, parameter::kmax=100
integer n
double precision a(n,n), b(n), x(n)
double precision c, omega, eps, delta, conv, sum
integer i, j, k, iter, flag

! check if the system is diagonally dominant
flag = 0
do i=1,n

sum = 0.0
do j=1,n

if(i == j) cycle
sum = sum+abs(a(i,j))

end do
if(abs(a(i,i)) < sum) flag = flag+1

end do
if(flag >0) write(*,*) ’The system is NOT diagonally dominant’

do k=1,kmax
conv = 0.0
do i=1,n

delta = b(i)
do j=1,n
delta = delta - a(i,j)*x(j)

end do
x(i) = x(i)+omega*delta/a(i,i)
if(abs(delta) > conv) conv=abs(delta)

end do
if(conv < eps) exit

end do
iter = k
if(k == kmax) write (*,*)’The system failed to converge’

end subroutine gs_sor

Example 6.5 Solution by successive-over-relaxation

The successive-over-relaxation (SOR)

Matrix A and vector b

8.00000 2.00000 4.00000 2.00000
2.00000 6.00000 1.00000 6.00000
1.00000 1.00000 8.00000 4.00000

Trial solutions x(n)
0.00000 0.00000 0.00000

Solutions x(n)
-0.20000 1.00000 0.40000

iterations = 10

Consider here to have an example from PDE
The Jacobi and Gauss-Seidel iterative methods are one step iterative methods since the xk+1

i

solution is defined through xk
i . In multi-step iterative methods, the xk+1

i solution is determined
in accordance with past iterations xk+1

i = f(xk
i , x

k−1
i , . . . , xk−m

i ).
There are multiple variations or iterative methods, like explicit and implicit iterative methods,

the method of upper relaxation, ... more?

6.4 Practical notes

There are multiple methods, and various computer packages available for solving systems of
linear equations. A researcher (or a student) faces this question - what would be a good way to
solve my problem? Should I invest my time in writing a program, buy software that could do
this job for me, learn how to use a sophisticated numerical package, or attempt to find a matrix
calculator on the Web?

The answer depends on the following factors: a) the complexity of the system of equations
(the size, conditioning, a general or sparse matrix), b) whether the problem is a part of a larger
computational problem or a standing alone task, c) whether a one-time solution is needed, or
multiple systems are to be solved.

A simple student problem can instantly be solved even with Excel. Excel has a number of
functions to work with matrices, in particularly MINVERSE to find an inverse matrix, and MMULT

for matrix multiplication. With Excel a solution can be just a few clicks away using x = A−1b.
Software packages such as Mathematica, Maple, or MathCad have libraries for solving various
systems of equations. If the problem is part of a larger computational project, and a system of
equations is not very large (less than a few hundreds of equation), yet well-conditioned, then
using the quick and efficient programs of this chapter would be best. However, for serious compu-
tational projects, it is advisable to use sophisticated packages developed by experts in numerical
analysis. The most well known commercial general libraries are NAG (Numerical Algorithmic
Group), and IMSL (International Mathematical and Statistical Library), both available in For-
tran 90/5 and C/C++. The NAG package also includes libraries for parallel calculations. The

IMSL library now is a part of compilers such as Intel Fortran, and Intel C++ (check it!).
Additionally, there are also various special packages to solve multiple problems of linear alge-

bra that are absolutely free. LAPACK (Linear Algebra PACKage) is the most advanced linear
algebra library. It provides routines for solving systems of linear equations and eigenvalue prob-
lems. LAPACK was originally written in Fortran 77, and was the successor of LINPAC (routines
for the linear equations) and EISPACK (set of routines for solving the eigenvalue problem). LA-
PACK has routines to handle both real and complex matrices in single and double precision.
The present core version of LAPACK is written in Fortran 90. It has several implementations:
LAPACK95 uses features of Fortran 95, CLAPACK in for C, LAPACK++ for C++ (it is being
superseded by the Template Numerical Toolkit (TNT), JLAPACK for Java.

There are also two large numerical libraries that have multiple routines for linear algebra
problems. SLATEC is a comprehensive library of routines having over 1400 general purpose
mathematical and statistical programs. The code was developed by a group from few National
Laboratories (US), and is therefore public domain. The library was written in Fortran 77, but
some routines are translated to Fortran 90, and there is a possibility to use SLATEC routines
from a C++ code. The other large library is the GNU Scientific Library (or GSL). It is written
in the C. The GSL is part of the GNU project and is distributed under the GNU General Public
License. GAMS - Guide to Available Mathematical Software from the National Institute of
Standards and Technology (NIST) is a practical starting point to find the best routine for your
problem. GAMS provides an excellent reference place and orientation for available programs.

6.5 Problems

1. Modify the Gauss 2 program above to calculate the determinant of a matrix A.

2. Using the routines from this chapter, write a program that evaluates the conditional num-
ber of a linear system (place eq number)

3. Modify the GS SOR program above based on Gauss-Seidel successive over-relaxation, to
change the convergence condition from (6.53) to (6.54).

4. Study on a diagonally dominant linear system how the choice of the factor ω affects the
convergence of the solution.

5. Implement a program from the LAPACK library to solve a system of linear equations(select
one or two)

6. Calculations: Compare accuracy of the program implementing the Gauss elimination
method with a program from a standard library for solutions of the following system
of equations⎛

⎜⎜⎜⎝
1 1

2
1
3

1
4

1
2

1
3

1
4

1

1

1

1
5

1
6

1

1

1
6

1
7

⎞
⎟⎟⎟⎠
⎛
⎜⎜⎜⎝

x1

x2

x3

x4

⎞
⎟⎟⎟⎠ =

⎛
⎜⎜⎜⎝

4
3
2
1

⎞
⎟⎟⎟⎠

7. consider some physics problems


